

# Treehouse Fastlane Audit Report

Dec 2, 2024





# Table of Contents

<b>Summary</b>	2
<b>Overview</b>	3
<b>Issues</b>	4
[WP-M1] Inverted condition for zero address check	4
[WP-M2] The unit of <code>getRedeemableAmount</code> should be in assets, but <code>totalRedeeming</code> 's <code>_earmark</code> is in shares	5
[WP-L3] The implementation (equivalent to require <code>_newFee &lt;= 501</code> ) is inconsistent with the NatSpec documentation's stated expectation of "fees (max 5%)"	7
[WP-L4] Modifying <code>waitingPeriod</code> affects all pending redemption requests, not just the redemption requests after the modification.	9
[WP-G5] Redundant <code>whenNotPaused</code> modifier in <code>redeemAndFinalize()</code>	11
[WP-I6] <code>finalizeRedeem</code> can be frontrun when the size of the redeem is large enough.	13
[WP-I7] <code>minRedeemInEth</code> can be misleading/prone to error when TASSET's <code>decimals</code> is not 18.	15
[WP-N8] <code>Unauthorized</code> and <code>RedemptionError</code> are defined but never used	17
[WP-N9] Unconventional event parameter naming style	18
<b>Appendix</b>	19
<b>Disclaimer</b>	20

## Summary

This report has been prepared for Treehouse smart contract, to discover issues and vulnerabilities in the source code of their Smart Contract as well as any contract dependencies that were not part of an officially recognized library. A comprehensive examination has been performed, utilizing Static Analysis and Manual Review techniques.

The auditing process pays special attention to the following considerations:

- Testing the smart contracts against both common and uncommon attack vectors.
- Assessing the codebase to ensure compliance with current best practices and industry standards.
- Ensuring contract logic meets the specifications and intentions of the client.
- Cross referencing contract structure and implementation against similar smart contracts produced by industry leaders.
- Thorough line-by-line manual review of the entire codebase by industry experts.

# Overview

## Project Summary

Project Name	Treehouse
Codebase	<a href="https://github.com/treehouse-gaia/tETH-protocol">https://github.com/treehouse-gaia/tETH-protocol</a>
Commit	f72c946322cbf56ad73a89f862262a156b8111cf
Language	Solidity

## Audit Summary

Delivery Date	Dec 2, 2024
Audit Methodology	Static Analysis, Manual Review
Total Issues	9

## [WP-M1] Inverted condition for zero address check

Medium

### Issue Description

In the `setFeeContract` function, the check for a zero address is incorrectly implemented.

The condition `_newContract != IFastlaneFee(address(0))` should be `_newContract == IFastlaneFee(address(0))`.

```
104     function setFeeContract(IFastlaneFee _newContract) external onlyOwner {
105         if (_newContract != IFastlaneFee(address(0))) revert ZeroAddress();
106         emit FeeContractUpdated(_newContract, feeContract);
107         feeContract = _newContract;
108     }
```

### Status

✓ Fixed

## [WP-M2] The unit of `getRedeemableAmount` should be in assets, but `totalRedeeming`'s `_earmark` is in shares

Medium

### Issue Description

The function `getRedeemableAmount()` attempts to calculate `_totalRedeemable` by subtracting two values with inconsistent units.

The issue is that `_earmark` is denominated in shares while `_underlyingInVault` is in assets.

```

114 function getRedeemableAmount() public view returns (uint _totalRedeemable) {
115     uint _underlyingInVault = IERC20(UNDERLYING).balanceOf(address(VAULT));
116     uint _earmark = ITreehouseRedemption(REDEMPTION_CONTRACT).totalRedeeming();
117
118     _totalRedeemable = _earmark > _underlyingInVault ? 0 : _underlyingInVault -
    _earmark;
119 }

```

```

81 function redeem(uint96 _shares) external nonReentrant whenNotPaused {
82     uint128 _assets = IERC4626(TASSET).previewRedeem(_shares).toUint128();
83     if (_assets < minRedeemInEth) revert MinimumNotMet();
84
85     IERC20(TASSET).safeTransferFrom(msg.sender, address(this), _shares);
86
87     redemptionInfo[msg.sender].push(
88         RedemptionInfo({
89             startTime: block.timestamp.toUint64(),
90             assets: _assets,
91             shares: _shares,
92             baseRate: _getBaseRate().toUint128()
93         })
94     );
95
96     unchecked {
97         redeeming[msg.sender] += _shares;
98         totalRedeeming += _shares;
99     }

```

```
100
101     emit Redeemed(msg.sender, _shares, _assets);
102 }
```

<https://github.com/treehouse-gaia/tETH-protocol/blob/40812a5bb857fa027c962bbfa55ae214356c039f/contracts/TreehouseFastlane.sol#L72-L89>

```
72  function redeemAndFinalize(uint96 _shares) external nonReentrant whenNotPaused {
73      uint _assets = IERC4626(TASSET).previewRedeem(_shares);
74      if (_assets < minRedeemInUnderlying) revert MinimumNotMet();
75      if (getRedeemableAmount() < _assets) revert InsufficientFundsInVault();
76
77      IERC20(TASSET).safeTransferFrom(msg.sender, address(this), _shares);
78      _assets = IERC4626(TASSET).redeem(_shares, address(this), address(this));
79      uint _fee = feeContract.applyFee(_assets);
80
81      IInternalAccountingUnit(IAU).burn(_assets);
82
83      REDEMPTION_CONTROLLER.redeem(_assets - _fee, msg.sender);
84      REDEMPTION_CONTROLLER.redeem(_fee, treasury);
85      // UNDERLYING.safeTransferFrom(address(VAULT), msg.sender, _assets - _fee);
86      // UNDERLYING.safeTransferFrom(address(VAULT), treasury, _fee);
87
88      emit Redeemed(msg.sender, _shares, _assets, _fee);
89 }
```

## Status

✓ Fixed

## [WP-L3] The implementation (equivalent to require \_newFee <= 501) is inconsistent with the NatSpec documentation's stated expectation of "fees (max 5%)"

Low

### Issue Description

501 (5.01%) is now allowed.

The expected max fee should be `500` instead of `501` .

```

12  /**
13   * @notice Fastlane fee contract
14   */
15  contract FastlaneFee is IFastlaneFee, Ownable2Step {
16      uint constant PRECISION = 1e4;
17      uint public fee = 200; // 2% in bips
18  }
19  @dev 18,35
20
21  /**
22   * Set fees (max 5%)
23   * @dev onlyOwner
24   * @param _newFee new fee in bips
25   */
26  function setFee(uint _newFee) external onlyOwner {
27      if (_newFee > 501) revert MaxFeeExceeded(); // max out at 5%;
28      emit FeeUpdated(_newFee, fee);
29      fee = _newFee;
30  }
31  }

```

### Recommendation

Consider changing to:

```

36  /**
37   * Set fees (max 5%)
38   * @dev onlyOwner

```



```
39     * @param _newFee new fee in bips
40     */
41     function setFee(uint _newFee) external onlyOwner {
42         if (_newFee > 500) revert MaxFeeExceeded(); // max out at 5%;
43         emit FeeUpdated(_newFee, fee);
44         fee = _newFee;
45     }
```

## Status

✓ Fixed

## [WP-L4] Modifying `waitingPeriod` affects all pending redemption requests, not just the redemption requests after the modification.

Low

### Issue Description

An alternative design would be to store the maturity time in the redeeming struct so that `setWaitingPeriod` only affects future redemption requests.

```

104  /**
105   * @notice Finalize tAsset redemption
106   * @param _redeemIndex index to finalize
107   */
108  function finalizeRedeem(
109      uint _redeemIndex
110  ) external nonReentrant whenNotPaused validateRedeem(msg.sender, _redeemIndex) {
111      RedemptionInfo storage _redeem = redemptionInfo[msg.sender][_redeemIndex];
112
113      if (block.timestamp < _redeem.startTime + waitingPeriod) revert
114      InWaitingPeriod();
115      uint _assets = IERC4626(TASSET).redeem(_redeem.shares, address(this),
116      address(this));
117      redeeming[msg.sender] -= _redeem.shares;
118      totalRedeeming -= _redeem.shares;
119
120      address _underlying = VAULT.getUnderlying();
121
122      uint _returnAmount = _getReturnAmount(_redeem.assets, _redeem.baseRate,
123      _assets, _getBaseRate());
124      uint _fee = (_returnAmount * redemptionFee) / PRECISION;
125      _returnAmount = _returnAmount - _fee;
126
127      if (_returnAmount > _redeem.assets) revert RedemptionError();
128
129      if (IERC20(_underlying).balanceOf(address(VAULT)) < _returnAmount) revert
130      InsufficientFundsInVault();
131      IInternalAccountingUnit(IAU).burn(_returnAmount);
132      REDEMPTION_CONTROLLER.redeem(_returnAmount, msg.sender);

```

```
129
130     // reused assignment - transfer leftover asset back into 4626
131     _assets = IERC20(IAU).balanceOf(address(this));
132
133     if (_assets > 0) {
134         IERC20(IAU).safeTransfer(TASSET, _assets);
135     }
136
137     emit RedeemFinalized(msg.sender, _returnAmount, _fee);
138
139     // last because deletes are in-place
140     _deleteRedeemEntry(_redeemIndex);
141 }
142
143 /**
144  * @notice Set the waiting period for finalizing redeems
145  * @param _newWaitingPeriod new waiting period in seconds
146  */
147 function setWaitingPeriod(uint32 _newWaitingPeriod) external onlyOwner {
148     emit WaitingPeriodUpdated(_newWaitingPeriod, waitingPeriod);
149     waitingPeriod = _newWaitingPeriod;
150 }
```

## Status

 Acknowledged

## [WP-G5] Redundant `whenNotPaused` modifier in `redeemAndFinalize()`

### Gas

### Issue Description

The `REDEMPTION_CONTROLLER.redeem()` at `TreehouseFastlane.sol` L83-84 already includes a `whenNotPaused` modifier, and `TreehouseFastlane` shares the same `paused` status with `REDEMPTION_CONTROLLER`.

```

68  /**
69   * @notice Atomically redeem tAsset
70   * @param _shares amount of tAsset to redeem
71   */
72  function redeemAndFinalize(uint96 _shares) external nonReentrant whenNotPaused {
73      uint _assets = IERC4626(TASSET).previewRedeem(_shares);
74      if (_assets < minRedeemInUnderlying) revert MinimumNotMet();
75      if (getRedeemableAmount() < _assets) revert InsufficientFundsInVault();
76
77      IERC20(TASSET).safeTransferFrom(msg.sender, address(this), _shares);
78      _assets = IERC4626(TASSET).redeem(_shares, address(this), address(this));
79      uint _fee = feeContract.applyFee(_assets);
80
81      IInternalAccountingUnit(IAU).burn(_assets);
82
83      REDEMPTION_CONTROLLER.redeem(_assets - _fee, msg.sender);
84      REDEMPTION_CONTROLLER.redeem(_fee, treasury);
85      // UNDERLYING.safeTransferFrom(address(VAULT), msg.sender, _assets - _fee);
86      // UNDERLYING.safeTransferFrom(address(VAULT), treasury, _fee);
87
88      emit Redeemed(msg.sender, _shares, _assets, _fee);
89  }

```

```

131  /**
132   * Inherits pause state from RedemptionController
133   */
134  function paused() public view override returns (bool) {
135      return REDEMPTION_CONTROLLER.paused();

```

```
136     }
```

```
42     /**
43      * Redeem underlying from vault to `_recipient`
44      * @dev only redemption contracts
45      * @param _amount amount to redeem
46      * @param _recipient recipient
47      */
48     function redeem(uint _amount, address _recipient) external whenNotPaused {
49         if (_redemptionContracts.contains(msg.sender) == false) revert Unauthorized();
50         IERC20(UNDERLYING).safeTransferFrom(address(VAULT), _recipient, _amount);
51     }
```

## Recommendation

Recommend removing the `Pausable` parent contract inheritance from `TreehouseFastlane` .

## Status

ⓘ Acknowledged

## [WP-I6] `finalizeRedeem` can be frontrun when the size of the redeem is large enough.

### Informational

### Issue Description

`finalizeRedeem` returns both the fee and share price difference back to the 4626; a huge redeem can cause a surge in share price.

If the sudden increase in share price ratio exceeds the Fastlane fee, attackers can profit by front-running the `finalizeRedeem` transaction and then back-running the Fastlane Redeem.

Given the fact that Fastlane has a significant fee, it's quite hard in practice for the frontrun to be profitable.

```


108 function finalizeRedeem(
109     uint _redeemIndex
110 ) external nonReentrant whenNotPaused validateRedeem(msg.sender, _redeemIndex) {
111     RedemptionInfo storage _redeem = redemptionInfo[msg.sender][_redeemIndex];
112
113     if (block.timestamp < _redeem.startTime + waitingPeriod) revert
    InWaitingPeriod();
114     uint _assets = IERC4626(TASSET).redeem(_redeem.shares, address(this),
    address(this));
115     redeeming[msg.sender] -= _redeem.shares;
116     totalRedeeming -= _redeem.shares;
117
118     address _underlying = VAULT.getUnderlying();
119
120     uint _returnAmount = _getReturnAmount(_redeem.assets, _redeem.baseRate,
    _assets, _getBaseRate());
121     uint _fee = (_returnAmount * redemptionFee) / PRECISION;
122     _returnAmount = _returnAmount - _fee;
123
124     if (_returnAmount > _redeem.assets) revert RedemptionError();
125
126     if (IERC20(_underlying).balanceOf(address(VAULT)) < _returnAmount) revert
    InsufficientFundsInVault();
127     IInternalAccountingUnit(IAU).burn(_returnAmount);

```

```
128     REDEMPTION_CONTROLLER.redeem(_returnAmount, msg.sender);
129
130     // reused assignment - transfer leftover asset back into 4626
131     _assets = IERC20(IAU).balanceOf(address(this));
132
133     if (_assets > 0) {
134         IERC20(IAU).safeTransfer(TASSET, _assets);
135     }
136
137     emit RedeemFinalized(msg.sender, _returnAmount, _fee);
138
139     // last because deletes are in-place
140     _deleteRedeemEntry(_redeemIndex);
141 }
```

```
17     uint public fee = 200; // 2% in bips
```

## Status

 Acknowledged

## [WP-I7] `minRedeemInEth` can be misleading/prone to error when TASSET's `decimals` is not 18.

### Informational

### Issue Description

For example, let's say the asset is USDC and the TASSET is tUSDC with a `decimals` of 8; then `minRedeemInEth` would be  $1e10$  times bigger than the face value.

```

81     function redeem(uint96 _shares) external nonReentrant whenNotPaused {
82         uint128 _assets = IERC4626(TASSET).previewRedeem(_shares).toUint128();
83         if (_assets < minRedeemInEth) revert MinimumNotMet();
84
85         IERC20(TASSET).safeTransferFrom(msg.sender, address(this), _shares);
86
87         redemptionInfo[msg.sender].push(
88             RedemptionInfo({
89                 startTime: block.timestamp.toUint64(),
90                 assets: _assets,
91                 shares: _shares,
92                 baseRate: _getBaseRate().toUint128()
93             })
94         );
95
96         unchecked {
97             redeeming[msg.sender] += _shares;
98             totalRedeeming += _shares;
99         }
100
101         emit Redeemed(msg.sender, _shares, _assets);
102     }

```

```

91     /**
92      * @notice Set the mininum redemption size
93      * @param _newMinRedeemInUnderlying new minimum in 1e18
94      */
95     function setMinRedeem(uint96 _newMinRedeemInUnderlying) external onlyOwner {
96         emit MinRedeemUpdated(_newMinRedeemInUnderlying, minRedeemInUnderlying);

```



```
97     minRedeemInUnderlying = _newMinRedeemInUnderlying;  
98   }  
99
```

## Status

✓ Fixed

## [WP-N8] Unauthorized and RedemptionError are defined but never used

### Issue Description

TreehouseRedemptionV2.sol

```
19 error Unauthorized();
```

```
20 error RedemptionError();
```

### Status

✓ Fixed

## [WP-N9] Unconventional event parameter naming style

### Issue Description

Event parameter names typically don't start with `_`.

When accessing event data, using parameter names that start with `_` becomes necessary, which is not ideal.

TreehouseFastlane.sol

```
17 interface ITreehouseFastlane {
18     error MinimumNotMet();
19     error InsufficientFundsInVault();
20     error RedemptionError();
21     error ZeroAddress();
22
23     event Redeemed(address indexed _user, uint _shares, uint _assets, uint _fee);
24     event MinRedeemUpdated(uint128 _new, uint128 _old);
25     event FeeContractUpdated(IFastlaneFee _new, IFastlaneFee _old);
26 }
```

### Status

✓ Fixed



# Appendix

## Timeliness of content

The content contained in the report is current as of the date appearing on the report and is subject to change without notice, unless indicated otherwise by WatchPug; however, WatchPug does not guarantee or warrant the accuracy, timeliness, or completeness of any report you access using the internet or other means, and assumes no obligation to update any information following publication.

## Disclaimer

This report is based on the scope of materials and documentation provided for a limited review at the time provided. Results may not be complete nor inclusive of all vulnerabilities. The review and this report are provided on an as-is, where-is, and as-available basis. You agree that your access and/or use, including but not limited to any associated services, products, protocols, platforms, content, and materials, will be at your sole risk. Smart Contract technology remains under development and is subject to unknown risks and flaws. The review does not extend to the compiler layer, or any other areas beyond the programming language, or other programming aspects that could present security risks. A report does not indicate the endorsement of any particular project or team, nor guarantee its security. No third party should rely on the reports in any way, including for the purpose of making any decisions to buy or sell a product, service or any other asset. To the fullest extent permitted by law, we disclaim all warranties, expressed or implied, in connection with this report, its content, and the related services and products and your use thereof, including, without limitation, the implied warranties of merchantability, fitness for a particular purpose, and non-infringement. We do not warrant, endorse, guarantee, or assume responsibility for any product or service advertised or offered by a third party through the product, any open source or third-party software, code, libraries, materials, or information linked to, called by, referenced by or accessible through the report, its content, and the related services and products, any hyperlinked websites, any websites or mobile applications appearing on any advertising, and we will not be a party to or in any way be responsible for monitoring any transaction between you and any third-party providers of products or services. As with the purchase or use of a product or service through any medium or in any environment, you should use your best judgment and exercise caution where appropriate. FOR AVOIDANCE OF DOUBT, THE REPORT, ITS CONTENT, ACCESS, AND/OR USAGE THEREOF, INCLUDING ANY ASSOCIATED SERVICES OR MATERIALS, SHALL NOT BE CONSIDERED OR RELIED UPON AS ANY FORM OF FINANCIAL, INVESTMENT, TAX, LEGAL, REGULATORY, OR OTHER ADVICE.